

# XFont Lib for FreeBASIC

## Introduction

XFont is Object Oriented Programming (OOP) library designed for drawing smooth font text. This lib support 1, 2, 4, 8, 15/16, 24/32 bpp depth.

Some of the most exciting features:

- Support alpha blending. This works in 15/16 and 24/32 bpp depths.
- Alpha mask for drawing smooth font text.
- 8 fonts slot can be loaded with x-font file format.
- Included with internal standard font.
- Easy to use.
- No initialize needed between switching screen mode.
- No ScreenLock/ScreenUnlock switch needed (For FB 0.20.0b or above only).
- Unicode support (ISO 10646).

## About

XFont Lib was developed by Victor Phoa.

Copyright 2008-2010 © Xaviorsoft Studios. All Rights Reserved.

## Note

All example code written is designed for FB 0.20.0 or above.

FB 0.18.x user should add ScreenLock/ScreenUnlock switch manually to make example code working correctly.

## Keywords List

(Arrange by mostly used order)

## DrawString

### Syntax

Declare Sub DrawString Overload (ByVal Target As Any Ptr=0, ByRef Text As wString Ptr, ByVal X As Short, \_  
ByVal Y As Short, ByVal ScaleX As Single=1, ByVal ScaleY As Single=1, \_  
ByVal Direction As uByte=0)

Declare Sub DrawString Overload (ByVal Target As Any Ptr=0, ByRef Text As String, ByVal X As Short, \_  
ByVal Y As Short, ByVal ScaleX As Single=1, ByVal ScaleY As Single=1, \_  
ByVal Direction As uByte=0)

Declare Sub DrawString Overload (ByVal Target As Any Ptr=0, ByRef Text As uInteger Ptr, ByVal X As Short, \_  
ByVal Y As Short, ByVal ScaleX As Single=1, ByVal ScaleY As Single=1, \_  
ByVal Direction As uByte=0)

### Usage

**Drawstring** [*target*], *text*, *x*, *y*, [*scalex*], [*scaley*], [*direction*]

### Description

Displays text at *x*, *y* in graphics modes. *target* specifies buffer to draw on. *target* may be an image created with ImageCreate or Get (Graphics). If omitted or 0, *target* defaults to the screen's current work page.

*x*, *y* are the coordinates of the pixel.

*scalex*, *scaley* are character magnification factor

*direction* are text orient style. If omitted or 0, text orient left to right. If 1, text orient bottom to top.

Is possible to draw unicode characters in DOS port using uInteger Pointer as long the end of the block memory is marked by 0.

Note : Maximum text were limited to 32767 characters.

### Example

```
#Include Once "xfont.bi"

Dim Shared As xfont.interface font
ScreenRes 640,480,32
font.drawstring ("Hello World!",10,10)
font.drawstring ("Hello World!",10,50,2,2) '2x size
font.drawstring ("Hello World!",10,200,,1) 'Vertical
Sleep
```

```
'Draw characters using uinteger ptr
#include Once "xfont.bi"
Dim as uinteger s(12) => {72, 101, 108, 108, 111, 32, 87, 111, 114,
108, 100, 33, 0}
Dim Shared As xfont.interface font
ScreenRes 640,480,32
font.drawstring (,@s(0),10,10) 'Draw "Hello World!"
Sleep
```

## LoadFont

### Syntax

Declare Function LoadFont Overload (Byref Filename As String, ByVal FontIndex As uByte) As Integer

Declare Function LoadFont Overload (ByRef DataPtr As uByte Ptr, ByVal FontIndex As uByte) As Integer

### Usage

*result* = **LoadFont** (*Filename*, *FontIndex*)

*result* = **LoadFont** (*Dataptr*, *FontIndex*)

### Description

Load font from file or memory location to empty font slot.

Zero is returned if LoadFont completed successfully.

### Example

```
'Load font from file
#include Once "xfont.bi"
Dim Shared As xfont.interface font
font.loadfont ("myfont.xf", 1) 'Load to FontIndex 1
font.fontindex = 1 'Use the font
ScreenRes 640,480,32
font.drawstring (,"Hello World!",10,10)
Sleep
```

```
'Load font from memory location
#include Once "xfont.bi"
Dim pmem As uByte Ptr
Dim As Integer f
Dim As LongInt l
Dim As String s = "myfont.xf"
Dim Shared As xfont.interface font

l=filelen(s)
pmem = Allocate(l*Len(uByte))
f = FreeFile
Open s For Binary As #f
Get #f,, *pmem, l
Close #f

font.loadfont (pmem, 1) 'Load to FontIndex 1
Deallocate pmem
font.fontindex = 1 'Use the font
ScreenRes 640,480,32
font.drawstring (,"Hello World!",10,10)
Sleep
```

## UnloadFont

### Syntax

Declare Function UnloadFont (ByVal FontIndex As uByte) As Integer

### Usage

*result* = **UnloadFont** (*fontindex*)

### Description

Unload font from its slot. Note: default font in slot 0 by default can't be unloaded.

Zero is returned if UnloadFont completed successfully.

### Example

```
#Include Once "xfont.bi"
Dim Shared As xfont.interface font
font.loadfont ("myfont.xf", 1) 'Load to FontIndex 1
font.fontindex = 1 'Use the font
ScreenRes 640,480,32
font.drawstring ("This is loaded font",10,10)
font.unloadfont (1)
font.drawstring ("This is standard font",10,50)
Sleep
```

## StringWidth

### Syntax

Declare Function StringWidth Overload (ByRef Text As wString Ptr) As uShort  
Declare Function StringWidth Overload (ByRef Text As String) As uShort  
Declare Function StringWidth Overload (ByRef Text As uInteger Ptr) As uShort

### Usage

*result* = **StringWidth** (*text*)

### Description

Returns the width of the text (based on the current font) in pixels.

### Example

```
#Include Once "xfont.bi"  
Dim Shared As xfont.interface font  
Print "Width for 'Hello World' is " & _  
      font.stringwidth ("Hello World") & " pixels"  
Sleep
```

## **TextHeight**

### Syntax

Declare Property TextHeight () As uShort

### Usage

*result* = **TextHeight**

### Description

Contains the height of a line of text drawn with the current font.

### Example

```
#Include Once "xfont.bi"  
Dim Shared As xfont.interface font  
Print "Height for text is " & font.textheight & " pixels"  
Sleep
```

## FontIndex

### Syntax

Declare Property FontIndex () As uByte  
Declare Property FontIndex (ByVal Expression As uByte)

### Description

Used to get and set the current font.

### Example

```
#Include Once "xfont.bi"  
Dim Shared As xfont.interface font  
font.loadfont ("myfont.xf", 1) 'Load to FontIndex 1  
ScreenRes 640,480,32  
font.drawstring (,"Use Fontindex " & font.fontindex,10,10)  
font.fontindex = 1  
font.drawstring (,"Use Fontindex " & font.fontindex,10,50)  
Sleep
```



## TextSize

### Syntax

Declare Property TextSize () As uShort

### Usage

*result* = **TextSize**

### Description

Used to get the current font size.

### Example

```
#Include Once "xfont.bi"  
Dim Shared As xfont.interface font  
Print "Current font size is " & font.textsize  
Sleep
```

## BackColor

### Syntax

Declare Property BackColor () As uInteger  
Declare Property BackColor (ByVal Expression As uInteger)

### Description

Used to get and set the display background color.

For 15/16, 24/32 bpp mode, bgcolor use AAAAAAARRRRRRRRRGGGGGGGGBBBBBBBB as color format.  
Where valid value for each color component is 0 to 31 for 15/16 bpp, and 0 to 255 for 24/32 bpp.

For 1, 2, 4 and 8 bpp mode, bgcolor use AAAAAAAXXXXXXXXXXXXXXXXXXPPPPPPP as color format.  
Where AAAAAA is background masking flag, and PPPPPPP is palette color index.

Set AAAAAA value < 128 will passed background drawing.

Tips : To make background appear is possible to set bgcolor expression with *palette or &hFF000000*.

Note : The first default bgcolor is set to &hFFFFFF

### Example

```
'Background color for non palette mode
#include Once "xfont.bi"
Dim Shared As xfont.interface font
ScreenRes 640,480,32
font.backcolor = Rgba(0,255,0,125)
font.drawstring ("Hello world",10,10)
Sleep
```

```
'Background color for palette mode
#include Once "xfont.bi"
Dim Shared As xfont.interface font
ScreenRes 640,480,8
font.backcolor = 2 Or &hff000000
font.drawstring ("Hello world",10,10)
Sleep
```

```
'Disable background for palette mode
#include Once "xfont.bi"
Dim Shared As xfont.interface font
ScreenRes 640,480,8
Color 0,15 : Cls
font.backcolor = 0
font.drawstring ("Hello world",10,10)
Sleep
```

## ForeColor

### Syntax

Declare Property ForeColor () As uInteger  
Declare Property ForeColor (ByVal Expression As uInteger)

### Description

Used to get and set the display foreground color.

For 15/16, 24/32 bpp mode, forecolor use AAAAAAARRRRRRRRRGGGGGGGBBBBBBBB as color format.  
Where valid value for each color component is 0 to 31 for 15/16 bpp, and 0 to 255 for 24/32 bpp.

For 1, 2, 4 and 8 bpp mode, forecolor use AAAAAAAXXXXXXXXXXXXXXXXXXPPPPPPPP as color format.  
Where AAAAAAA is foreground masking flag, and PPPPPPPP is palette color index.

Set AAAAAAA value < 128 will passed foreground drawing.

Tips : To make background appear is possible to set foreolor expression with *palette or &hFF000000*.

Note : The first default forecolor is set to &hFF000000

### Example

```
'foreground color for non palette mode
#include Once "xfont.bi"
Dim Shared As xfont.interface font
ScreenRes 640,480,32
font.forecolor = Rgba(0,255,0,255)
font.drawstring ("Hello world",10,10)
Sleep
```

```
'foreground color for palette mode
#include Once "xfont.bi"
Dim Shared As xfont.interface font
ScreenRes 640,480,8
font.forecolor = 2 Or &hff000000
font.drawstring ("Hello world",10,10)
Sleep
```

## BackColor

### Syntax

Declare Property BackColor () As uInteger  
Declare Property BackColor (ByVal Expression As uInteger)

### Description

Used to get and set the display background color.

For 15/16, 24/32 bpp mode, bgcolor use AAAAAAARRRRRRRRRGGGGGGGGBBBBBBBB as color format.  
Where valid value for each color component is 0 to 31 for 15/16 bpp, and 0 to 255 for 24/32 bpp.

For 1, 2, 4 and 8 bpp mode, bgcolor use AAAAAAAXXXXXXXXXXXXXXXXXXPPPPPPPP as color format.  
Where AAAAAAA is background masking flag, and PPPPPPP is palette color index.

Set AAAAAAA value < 128 will passed background drawing.

Tips : To make background appear is possible to set bgcolor expression with *palette or &hFF000000*.

Note : The first default bgcolor is set to &hFFFFFF

### Example

```
'Background color for non palette mode
#include Once "xfont.bi"
Dim Shared As xfont.interface font
ScreenRes 640,480,32
font.backcolor = Rgba(0,255,0,125)
font.drawstring ("Hello world",10,10)
Sleep
```

```
'Background color for palette mode
#include Once "xfont.bi"
Dim Shared As xfont.interface font
ScreenRes 640,480,8
font.backcolor = 2 Or &hff000000
font.drawstring ("Hello world",10,10)
Sleep
```

```
'Disable background for palette mode
#include Once "xfont.bi"
Dim Shared As xfont.interface font
ScreenRes 640,480,8
Color 0,15 : Cls
font.backcolor = 0
font.drawstring ("Hello world",10,10)
Sleep
```

## UseViewport

### Syntax

Declare Property UseViewport(ByVal Expression As Byte)

Declare Property UseViewport() As Byte

### Description

Used to get and set the viewport flag.

If set to non zero, clipping region (also known as viewport) will be used.

### Example

```
'Using viewport
#include Once "xfont.bi"
Dim Shared As xfont.interface font

ScreenRes 640,480,32
Color 0, &hFF00FF
Cls
font.UseViewport = -1
View (19,18)-(70,180)
font.drawstring ("Hello World",10,10)
font.drawstring ("Hello World",10,50,2,2) '2x size
font.drawstring ("Hello World",10,200,,1) 'Vertical
font.drawstring ("Hello World",50,200,2,2,1) 'Vertical
Sleep
```

## IsLoaded

### Syntax

Declare Property IsLoaded(ByVal Expression As uByte) As Byte

### Description

Used to get status if font was loaded into font slot.

-1 is returned if Font was loaded in *fontindex* slot.

### Usage

*result* = **IsLoaded**(*fontindex*)

### Example

```
'Using IsLoaded
#include Once "xfont.bi"
Dim Shared As xfont.interface font

If font.IsLoaded(1) Then
    Print "Font loaded"
Else
    Print "Font empty"
End If
Sleep
```

## FontName

### Syntax

Declare Property FontName () As String

### Usage

*result* = **FontName**

### Description

Used to get the current fontname.

### Example

```
#Include Once "xfont.bi"  
Dim Shared As xfont.interface font  
Print "Current fontname is " & font.fontname  
Sleep
```

## **Bold**

### Syntax

Declare Property Bold () As Byte

### Usage

*result* = **Bold**

### Description

Used to get the current font bold style flag.

### Example

```
#Include Once "xfont.bi"  
Dim Shared As xfont.interface font  
Print "Current font bold : " & font.bold  
Sleep
```



## **Italic**

### Syntax

Declare Property Italic () As Byte

### Usage

*result* = **Italic**

### Description

Used to get the current font italic style flag.

### Example

```
#Include Once "xfont.bi"  
Dim Shared As xfont.interface font  
Print "Current font italic : " & font.italic  
Sleep
```

## **Underline**

### Syntax

Declare Property Underline () As Byte

### Usage

*result* = **Underline**

### Description

Used to get the current font underline style flag.

### Example

```
#Include Once "xfont.bi"  
Dim Shared As xfont.interface font  
Print "Current font underline : " & font.underline  
Sleep
```

## Message

### Syntax

Declare Property Message () As String

### Usage

*result* = **Message**

### Description

Used to get the current font message.

### Example

```
#Include Once "xfont.bi"  
Dim Shared As xfont.interface font  
Print "Current font message : " & font.message  
Sleep
```

## **LastError**

### Syntax

Declare Property LastError () As Integer

### Usage

*result* = **LastError**

### Description

Contains the error code for the last supported operation.

The following error codes may be returned:

101 : FileNotFound

102 : IndexOutOfBounds

103 : InvalidFileSize

104 : InvalidSignature

105 : VersionMismatch

106 : InvalidHandle / Illegal Function Call

### Example

```
#Include Once "xfont.bi"  
Dim Shared As xfont.interface font  
font.loadfont ("?",1) 'generate some error  
Print "Last error : " & font.lasterror  
Sleep
```

## ClearError

### Syntax

Declare Sub ClearError ()

### Usage

**ClearError**

### Description

Clear the error code for the last supported operation.

### Example

```
#Include Once "xfont.bi"  
Dim Shared As xfont.interface font  
font.loadfont ("?",1) 'generate some error  
Print "Last error : " & font.lasterror  
font.clearerror 'clear error  
Print "Last error : " & font.lasterror  
Sleep
```

## MajorVersion

### Syntax

Declare Property MajorVersion () As uShort

### Usage

*result* = **MajorVersion**

### Description

Used to get Major version of xfont library.

### Example

```
#Include Once "xfont.bi"  
Dim Shared As xfont.interface font  
Print "Lib major version : " & font.majorversion  
Sleep
```

## MinorVersion

### Syntax

Declare Property MinorVersion () As uShort

### Usage

*result* = **MinorVersion**

### Description

Used to get Minor version of xfont library.

### Example

```
#Include Once "xfont.bi"  
Dim Shared As xfont.interface font  
Print "Lib minor version : " & font.minorversion  
Sleep
```

## Revision

### Syntax

Declare Property Revision () As uShort

### Usage

*result* = **Revision**

### Description

Used to get Revision of xfont library.

### Example

```
#Include Once "xfont.bi"  
Dim Shared As xfont.interface font  
Print "Lib revision : " & font.revision  
Sleep
```



## Copyright

### Syntax

Declare Property Copyright () As String

### Usage

*result* = **Copyright**

### Description

Used to get Copyright string of xfont library.

### Example

```
#Include Once "xfont.bi"  
Dim Shared As xfont.interface font  
Print font.copyright  
Sleep
```

End of document